

A Quick Introduction to OCaml

Lorenzo Ceragioli

September 26, 2024

IMT Lucca

Functional Languages

- Programs are constructed by applying and composing functions
- Functional Languages are declarative: function definitions are expressions (mapping values to values)
- Heavily based on expressions
- Roughly abstracting away from the memory
- Functions are as every other value: first-class citizens
- In contrast to imperative languages where
 - Expressions are only a small part of the constructs
 - The programmer defines and uses commands for updating the memory

Pure Functions

- In a *pure* functional language, functions are mathematical functions (i.e. same input implies same output)
- Functional languages are often enriched with *imperative features*, i.e. not all functions are mathematical functions.
- For example for generating random numbers
- We will mostly deal with the pure fragment of OCaml, please avoid imperative features unless they are really needed!

How to Program in a Pure Functional Languages

- Use recursion instead of loops
- Recall that functions are first class citizens: you can have them as input and outputs of your functions
- Lots of library functions are like this: use them!
- No assignments implies no value can be updated: you usually create a new value each time
- The compiler is smart enough to avoid copying the data structure and just updates it if it is legal

OCaml Basics

OCaml Read Evaluate Print Loop

```
1 user@machine:~$ utop
2 # 5 + 4;;      (* expression *)
3 - : int = 9    (* its type and value *)
4
5 # let s = "hello world";; (* expression definition *)
6 val s : string = "hello world"
7 (* its name, type and value *)
8
9 # print_endline s;;    (* command *)
10 hello world           (* executing it *)
11 - : unit = ()         (* returned type and value *)
```

Hello World and Comments

The program `helloworld.ml`:

```
1   let () = print_endline "Hello , World!"  
2   (* this is a multiline  
3   comment *)
```

Compiling it:

```
$ ocamlc -o main main.ml
```

Executing it:

```
$ ./main  
Hello , World!
```

... but it is simpler to use *dune*!

Some basic types ...

Type	Name	Values
Integer numbers	<code>int</code>	<code>1, 2, 3, ...</code>
Boolean	<code>bool</code>	<code>true, false</code>
String	<code>string</code>	<code>"a", "b", "aa", ...</code>
Unit type	<code>unit</code>	<code>()</code>

Note: There is no implicit conversion, e.g. from `int` to `string`

And operators

Integer arithmetic	+, -, *, /, mod	
Boolean operators	not, &&,	
Structural Equivalence	=, <>	
Physical Equivalence	==, !=	← avoid this
Comparison	<, >, <=, >=	
String append	^	

If-then-else

`if $expr_1$ then $expr_2$ else $expr_3$`

Note:

- Choose between expressions, not commands or blocks
- The `else` case is mandatory!
- $expr_1$ must evaluate to a boolean (strongly inferred types!)
- $expr_2$ and $expr_3$ must have the same (inferred) type

```
1 # if "a" = "a" then 42 else 17;;  
2 - : int = 42
```

let and let ... in

Defining names **globally** : `let name = expr`

Defining names **locally** : `let name = expr1 in expr2`

```
1 # let a =
2     let b = "ciao" in
3         if b <> "ciao" then 42 else 17;;
4 val a : int = 17
5
6 # print_int a;;
7 17
8 - : unit = ()
9
10 # print_string b;;
11 Error: Unbound value x
```

let is not an assignment

Note: you can redefine the value of a name based on the previous name, but it is not an assignment as in imperative languages

```
1 # let a = 4;;  
2  
3 # let get_a x = a;;  
4  
5 # let a = 4 + a;;  
6 val a : int = 8  
7  
8 # get_a 3;;  
9 - : int = 4
```

The name a is really just a name for a value!

Subsequent definitions update the names, not the values!

Simple Functions (finally having some fun!)

Functions are values in OCaml, i.e. fully evaluated expressions

```
1  # fun x -> x + 1;;
2  - : int -> int = <fun>  (* inferred type (no code) *)
3
4  # (fun x -> x + 1) 5;;  (* fun application *)
5  -: int = 6
6
7  # let inc = fun x -> x + 1;;  (* naming a function *)
8  val inc : int -> int = <fun>
9
10 # let inc x = x + 1;;  (* same as before *)
11 val inc : int -> int = <fun>
```

Mind the definition order!

Functions are defined in order

```
1 # let inc x = (dec x) + 2
2 let dec x = x - 1;;
3 Error: Unbound value dec
4
5 # let dec x = x - 1
6 let inc x = (dec x) + 2;;
7 val dec : int -> int = <fun>
8 val inc : int -> int = <fun>
```

Polymorphic Functions

In OCaml types are inferred, what if multiple types are possible?

```
1 # let id x = x
2   val id : 'a -> 'a = <fun>
```

- The identity function works with every possible type
- 'a is a type variable, it stands for any type
- Note that the return type is also any type, but it coincides with the input type
- We will see polymorphic types that are not "any type"

Recursive Functions

Really just use the `rec` keyword and you can use the function name inside the function code

```
1 # let rec bang x =           (* naming the function *)
2   if x = 0 then 1             (* base case *)
3   else x * (bang (x-1));;      (* recursion *)
4 val bang : int -> int = <fun>
```

Mind that recursive functions may diverge, e.g. `bang -1`

Mutually Recursive Functions

Recall: functions must be defined before they are used in the body of other functions

If you want `f` to be defined in terms of `g` and vice-versa, use `rec` and the `and` keywords

```
1 # let rec f x =                (* defines f *)
2   if x < 1 then true
3   else g (x-1)                (* uses g *)
4 and g x =                      (* defines g *)
5   if x < 1 then false
6   else f (x-1);;              (* uses f *)
7 val f : int -> bool = <fun>
8 val g : int -> bool = <fun>
```

What do `f` and `g` compute?

Simple modules: .ml and .mli files

Idea: Split the code into modules and to hide the implementation.

The interface define the visible values promised by the module

```
modulename.mli ← the interface
```

```
val inc : int -> int
```

The implementation must satisfy the promises

```
modulename.ml ← the implementation
```

```
let dec x = x - 1      (* not visible *)
```

```
let inc x = (dec x) + 2 (* visible *)
```

Accessing the module's function from outside

```
1 let two = Modulename.inc 1
```

Compiling Modules (better to use dune)

`modulename.mli` ← the interface

```
val inc : int -> int
```

`modulename.ml` ← the implementation

```
let dec x = x - 1      (* not visible *)  
let inc x = (dec x) + 2 (* visible *)
```

`main.ml` ← the program

```
let () = print_int Modulename.inc 1
```

```
$ ocamlc -c modulename.ml <— (creates modulename.cmi)  
$ ocamlc -c modulename.ml <— (creates modulename.cmo)  
$ ocamlc -c main.ml <— (creates main.cmo)  
$ ocamlc -o ex modulename.cmo main.cmo <— (creates ex)  
$ ./ex  
2
```

Types

Pairs

Ordered pairs of elements of possibly different types

```
1 # let a = (4,"ciao");;
2 val a : int * string = (4, "ciao")
3
4 # fst a;;
5 - : int = 4
6
7 # snd a;;
8 - : string = "ciao"
9
10 # fst;;
11 - : 'a * 'b -> 'a = <fun>
12
13 # let sum (x,y) = x + y;;
14 val sum : int * int -> int = <fun>
```

Tuples

Ordered collections of elements of possibly different types (pairs are a special case)

```
1 # let b = (1,5,(6,true));;
2 val b : int * int * (int * bool) = (1, 5, (6, true))
3
4 # fst b;;
5 Error: This expression has type int * int * (int * bool)
6       but an expression was expected of type 'a * 'b
7
8 # fst;;    (* fst and snd only defined on pairs *)
9 - : 'a * 'b -> 'a = <fun>
```

Note: the type depends on the length of the tuple!

Lists

```
1  # [];;          (* Empty list ... *)
2  - : 'a list = [] (* ... is polymorphic! *)
3
4  # [1];;
5  - : int list = [1]
6
7  # [1;2;3;4;5;6];;
8  - : int list = [1; 2; 3; 4; 5; 6]
9
10 # ["a";4];;
11 Error: This expression has type int but an expression
12    was expected of type string
```

Note: all elements must be of the same type

Note: the type do NOT depend on the length

Lists/Tuple confusion

```
1 # [(4,5);(6,7);(8,9)];;
2 - : (int * int) list = [(4, 5); (6, 7); (8, 9)]
3
4 # [(4,5)];;
5 - : (int * int) list = [(4, 5)]
6
7 # [4,5];;
8 - : (int * int) list = [(4, 5)]
9
10 # 4,5;;  (* because parenthesis are optional! *)
11 - : int * int = (4, 5)
12
13 # [4,5,6,7];;
14 - : (int * int * int * int) list = [(4, 5, 6, 7)]
```

Note: use **semicolon** for lists!

Operations on Lists

Operation	Example	Note
Cons	<code>4 :: [1;2;3] = [4;1;2;3]</code>	fast
Concat	<code>[4;5;6] @ [1;2;3] = [4;5;6;1;2;3]</code>	$O(n)$
Head	<code>List.hd [4;5;6] = 4</code>	fast
Tail	<code>List.tl [4;5;6] = [5;6]</code>	fast
Reverse	<code>List.rev [4;5;6] = [6;5;4]</code>	$O(n)$

Note: List is the module of lists, several useful functions other than `hd`, `tl`, and `rev`.

User Defined Types

- Users can define their own types with the `type` construct
- Types can be recursive (no need for `rec` keyword)
- Polymorphic types defined through *type parameters*
- Some examples follow

Records

Idea: like tuples but with names

```
1 # type point = { x : int; y : int; z : int };;
2 type point = { x : int; y : int; z : int; }
3
4 # let origin = { x = 0; y = 0; z = 0 };;
5 val origin : point = {x = 0; y = 0; z = 0}
6
7 (* a new point with some changed values *)
8 # let p = { origin with x = 10; y = 5 };;
9 val p : point = {x = 10; y = 5; z = 0}
10
11 # p.x;;
12 - : int = 10
```

Note: must be explicitly defined as types!

Variants

Elements can be of either of the given (named) shapes.

```
1 type qual_temp = Hot | Cold | Fine
2
3 type temp =
4     Precise of int
5     | Approx of qual_temp
```

- `qual_temp` is an *enumeration* of atomic values
- `temp` is either a `qual_temp` (tagged as `Approx`) or a precise `int` for the degrees (tagged as `Precise`)
- Tags must be globally unique and start with a capital letter

Arithmetic expressions

```
1 type a_exp =  
2   Aval of int  
3   | Plus of a_exp * a_exp  
4   | Minus of a_exp * a_exp  
5   | Times of a_exp * a_exp
```

- You can define `exp` in terms of `exp` itself
- This is useful for defining the abstract syntax trees of programming languages

Mutually Recursive Variants

Arithmetic and boolean expressions

```
1  type a_exp =  
2    Aval of int  
3    | Plus of a_exp * a_exp  
4    | Minus of a_exp * a_exp  
5    | Times of a_exp * a_exp  
6  
7  type b_exp =  
8    Bval of bool  
9    | And of b_exp * b_exp  
10   | Or of b_exp * b_exp  
11   | Not of b_exp  
12   | Minor of a_exp * a_exp
```

- Boolean expressions defined in terms of arithmetic expressions
- What if we want a boolean to be considered also an integer (e.g. false = 0 and true = 1 as in C)?

Mutually Recursive Variants

Arithmetic and boolean expressions

```
1  type a_exp =  
2    Aval of int  
3    | Plus of a_exp * a_exp  
4    | Minus of a_exp * a_exp  
5    | Times of a_exp * a_exp  
6    | Of_bool of bexp  
7  and b_exp =  
8    Bval of bool  
9    | And of b_exp * b_exp  
10   | Or of b_exp * b_exp  
11   | Not of b_exp  
12   | Minor of a_exp * a_exp
```

Note: we use `and` as in mutually recursive functions

Polymorphic Variants

Like for lists, we can define polymorphic types through type parameters

```
1 type 'a btree =  
2   Leaf of 'a  
3   | Node of 'a btree * 'a btree  
4  
5 let tree1 =  
6   Node (Leaf 1,  
7         Node (Leaf 2, Leaf 3))  
8   (* is of type int tree*)  
9  
10 let tree2 =  
11   Node (Leaf "a",  
12        Node (Leaf "b", Leaf "c"))  
13   (* is of type string tree*)
```


Pattern Matching

Pattern matching is a form of branching similar to `switch case`

```
1  let f x =  
2      match x with  
3      | 0  
4      | 1 -> "one or less"  
5      | 2  
6      | 3 -> "two or three"  
7      | _ -> "four or more"
```

Note: the wildcard `_` match everything

Pattern Matching Variants

Pattern matching can

- branch based on the syntactic structure (i.e. how the value is build according to the type definition)
- bind the values of subterms to local names

```
1  let rec count_leaves x =  
2    match x with  
3    | Leaf _ -> 1  
4      (* I don't care the value inside the leaf *)  
5    | Node (x, y) ->  
6      (* x and y are local names for the subterms *)  
7      (count_leaves x) + (count_leaves y)
```

Note: the compiler will complain if some case is missing

Function Syntax

Pattern matching can be used implicitly with the keyword `function`

```
1  let rec count_leaves x =  
2      match x with  
3      | Leaf _ -> 1  
4      | Node (x, y) ->  
5          (count_leaves x) + (count_leaves y)
```

Is the same as

```
1  let rec count_leaves = function  
2      | Leaf _ -> 1  
3      | Node (x, y) ->  
4          (count_leaves x) + (count_leaves y)
```

Pattern Matching Variants: `as` and `when`

Computing a list of *leaves* containing a value smaller than 10 (the type of the tree is automatically inferred)

```
1  let rec less_ten_leaves x =  
2    match x with  
3    | Leaf n as leaf when n < 10 -> [leaf]  
4      (* n is the name of the subterm *)  
5      (* leaf is the name of the whole term *)  
6      (* therms are matched only when n < 10 *)  
7    | Leaf _ -> []  
8      (* all the other kind of leaves, i.e. n >= 10 *)  
9    | Node (x, y) ->  
10      (less_ten_leaves x) @ (less_ten_leaves y)
```

Note: sometimes they make your code a lot easier to read!

Pattern Matching with list

The cons operator `::` is not actually an operator, it is a type constructor!

```
1 # (@) ;;
2 - : 'a list -> 'a list -> 'a list = <fun>
3
4 # (::) ;;
5 Error: The constructor :: expects 2 argument(s),
6         but is applied here to 0 argument(s)
```

Hence, we can use it for pattern matching!

Pattern Matching with list

Weird length of a list

```
1 let rec weird_length ls =  
2   match ls with  
3   | [] -> "zero"  
4   | [_; _] -> "two"  
5   | _ :: ls' -> "one plus " ^ (weird_length ls')  
6  
7 # weird_length [0;0;0;0;0;0];;  
8 - : string = "one plus one plus one plus one plus two"  
9  
10 # weird_length [0];;  
11 - : string = "one plus zero"
```

Exercise 1. Write a pair of functions for evaluating arithmetical and boolean expressions without the `Of_bool` case

Exercise 2. Same but with the `Of_bool` case

Exercise 3. Write a type for general polymorphic trees (i.e. with any number of children)

Exercise 4. Write a function that packs consecutive duplicates of the input list elements into sublists.

For example, the function over `[0;0;2;3;3;3;0;2]` must return `[[0;0];[2];[3;3;3];[0];[2]]`

More on Functions

Curried Functions

- Addition usually takes two parameters:

```
1 let sum (x,y) = x + y
2 val sum: int * int -> int
```

- In OCaml, one usually writes the curried version (also the kind of functions that you will find in the libraries!)

```
1 let sum x y = x + y
2 val sum: int -> int -> int = <fun>
```

- the latter is the same as writing

```
1 let sum x = fun y -> x + y
2 val sum: int -> int -> int = <fun>
```

Curried Functions

- Why curried functions? For allowing partial application!

```
1  let addtwo =  
2    let sum x y = x + y  
3    in  sum 2  
4  val sum: int -> int = <fun>  
5  
6  # addtwo 3  
7  - : int = 5
```

- Functions are first class citizens in OCaml, hence they can be the returned value of a function
- A function over functions is called an *higher order function*!

Higher Order Functions

Functions as inputs and/or outputs

```
1  let twice f x =  
2    f (f x)  
3  val twice: ('a -> 'a) -> 'a -> 'a = <fun>  
4  
5  # twice not;;  
6  - : bool -> bool = <fun>  
7  
8  let fun_pair f g =  
9    fun (x, y) -> (f x, g y)  
10 val fun_pair : ('a -> 'b) -> ('c -> 'd)  
11     -> 'a * 'c -> 'b * 'd = <fun>  
12  
13 # fun_pair not addtwo;;  
14 - : bool * int -> bool * int = <fun>
```

Useful Functions in the List module

`map : ('a -> 'b) -> 'a list -> 'b list`

`map f ls` applies the function `f` to each element of a list to produce another list.

```
1 # List.map (fun x -> x + 2) [0;2;4;6;8];;  
2 - : int list = [2; 4; 6; 8; 10]
```

Note: if you need the index too, use `List.mapi`

```
1 # List.mapi (fun i x -> i + x) [0;2;4;6;8];;  
2 (* same as List.mapi (+) [0;2;4;6;8];; *)  
3  
4 - : int list = [0; 3; 6; 9; 12]
```

Useful Functions in the List module

`filter : ('a -> bool) -> 'a list -> 'a list`

`filter f ls` returns a list containing the elements of `ls` that satisfies the function `f` (preserving the order).

```
1 # List.filter (fun x -> x mod 2 = 0) [0;3;4;6;9];;  
2 - : int list = [0; 4; 6]
```

Useful Functions in the List module

`fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a`

`fold_left f initial ls` uses each element of the list to update a value initially instantiated as `initial`; the value is updated using the given function `f`; the result is the one obtained when all the list has been used.

It is the same as `f (...(f (f initial b1) b2)...) bn` when `ls = [b1; b2 ; ...; bn]`.

```
1 # List.fold_left (fun acc x -> acc + x) 0 [0;1;2;3;4];;  
2 - : int = 10
```

Different Styles of Combining Functions

With explicit parameters

```
1 let sum_even ls =  
2   List.fold_left  
3     (fun acc x -> acc + x)  
4     0  
5     (List.filter  
6       (fun x -> x mod 2 = 0)  
7       ls)
```

Different Styles of Combining Functions

With intermediate results

```
1 let sum_even ls =  
2   let even =  
3     List.filter  
4       (fun x -> x mod 2 = 0)  
5       ls  
6   in List.fold_left  
7     (fun acc x -> acc + x)  
8     0  
9     even
```


Different Styles of Combining Functions

The (almost) mathematician composition: with the @@ operator

```
1 let sum_even ls =  
2   List.fold_left  
3     (fun acc x -> acc + x)  
4     0  
5   @@  
6   List.filter  
7     (fun x -> x mod 2 = 0)  
8   @@  
9   ls
```

Note: @@ stands for function application, because of associativity behaves almost as a function composition $(f \circ g)x = f(g\ x)$, but the input `ls` must be given

Different Styles of Combining Functions

The (almost) computer scientist composition: with the `|>` operator

```
1 let sum_even ls =  
2   ls  
3   |>  
4   List.filter  
5     (fun x -> x mod 2 = 0)  
6   |>  
7   List.fold_left  
8     (fun acc x -> acc + x)  
9     0
```

Note: `|>` stands for function pipelining, because of associativity behaves almost as a function composition $(f; g)x = g(f\ x)$, but the input `ls` must be given

Exercise 5. Write a function that given two functions f and g return a function over pairs defined as f on the first element and g on the second element.

Exercise 6. Write a function that given a list of integers l_1 returns a list l_2 of the same length such that each element of l_2 in position i is the sum of all the elements in l_1 with position less or equal to i .

E.g. The function over $[3,6,10,2]$ returns $[3,9,19,21]$

Exercise 7. Define a type for Finite Automata and a function for checking if a given string is inside the generated language.

Advanced Features

OCaml's imperative features (but you should use pure functions)

```
1 let count = ref 0  (* of type int ref *)
2
3 let get_inc () =  (* of type unit -> int *)
4   count := !count + 1;
5   !count
6
7 # get_inc();;
8 - : int = 1
9 # get_inc();;
10 - : int = 2
11 # count;;
12 - : int ref = {contents = 2}
13 # !count;;
14 - : int = 2
```

Note: also mutable record fields, arrays, for and while loops.

Nested Modules

You can define a nested module inside a file (modulename.ml), i.e. inside the module Modulename itself.

```
1 module Username : sig (* interface of the module *)
2   type t
3   val of_string : string -> t
4   val to_string : t -> string
5 end = struct (* implementation of the module *)
6   type t = string
7   let of_string x = x
8   let to_string x = x
9 end
```

Module Type and Modules

You can split the signature (interface) and its implementation

- Code hiding
- E.g. you can put the signature of `Username` in the interface of the outmost module `modulename.mli`

```
1 module type ID = sig
2   type t
3   val of_string : string -> t
4   val to_string : t -> string
5 end
6
7 module String_id : ID = struct
8   type t = string
9   let of_string x = x
10  let to_string x = x
11 end
```

Repetita: Accessing Module Values and Types

We can either qualify the names

```
1  module type ID = sig
2      type t
3      val of_string : string -> t
4      val to_string : t -> string
5  end
6
7  module String_id : ID = struct
8      type t = string
9      let of_string x = x
10     let to_string x = x
11 end
12
13 # String_id.of_string "ciao"
14 - : String_id.t = <abstr>
```


Repetita: Accessing Module Values and Types

Or we can open the module (everything is automatically qualified)

```
1  module type ID = sig
2    type t
3    val of_string : string -> t
4    val to_string  : t -> string
5  end
6
7  module String_id : ID = struct
8    type t = string
9    let of_string x = x
10   let to_string x = x
11 end
12
13 open String_id
14
15 # of_string "ciao"
16 - : String_id.t = <abstr>
```

Including a Module

`include`: copies the content of a module into the current one

```
1 module ID : sig
2   type t
3   val of_str : string -> t
4   val to_str : t -> string
5 end = struct
6   type t = string
7   let of_str x = x
8   let to_str x = x
9 end
```

```
1 module ExtID : sig
2   type t
3   val of_str : string -> t
4   val to_str : t -> string
5   val of_int : int -> t
6 end = struct
7   include ID
8   let of_int x = x
9     |> string_of_int
10    |> of_str
11 end
```

Including a Type Module

`include`: also copies the content of an interface into the current one

```
1 module type ID =
2   sig
3     type t
4     val of_str : string -> t
5     val to_str : t -> string
6   end
7
8 module StrID : ID =
9   struct
10     type t = string
11     let of_str x = x
12     let to_str x = x
13   end
```

```
1 module type ExtID =
2   sig
3     include ID
4     val of_int : int -> t
5   end
6
7 module ExtStrID : ExtID =
8   struct
9     include StrID
10    let of_int x = x
11      |> string_of_int
12      |> of_str
13  end
```

A module for ordered lists.

```
1  module type OrdList = sig
2      type elem
3      type t
4      val empty : t
5      val to_list : t -> elem list
6      val add : elem -> t -> t
7  end
8
9  module IntListOrdList : OrdList = struct
10     type elem = int
11     type t = elem list
12     let empty = []
13     let to_list ls = ls
14     let rec add l ls = match ls with
15         | [] -> [l]
16         | l'::ls' -> if (l <= l' ) then l::l'::ls'
17                     else l':: (add l ls')
18  end
```

Functors: map between modules

Build a module from one that satisfies a given interface!

```
1 module type TotOrd = sig
2   type t
3   val lesseq : t -> t -> bool
4 end
5
6 module MakeOrdList (Elem : TotOrd) :
7   (OrdList with type elem = Elem.t) = struct
8   type elem = Elem.t
9   type t = elem list
10  let empty = []
11  let to_list ls = ls
12  let rec add l ls = match ls with
13    | [] -> [l]
14    | l'::ls' -> if (Elem.lesseq l l')
15      then l::l'::ls' else l'::(add l ls')
16 end
```

Using Functors

```
1 module OrdInt : TotOrd = struct
2   type t = int
3   let lesseq n1 n2 = n1 <= n2
4 end
5
6 module IntOrdList : (OrdList with type elem = int) =
    MakeOrdList (OrdInt)
7
8 # IntOrdList.empty
9 |> IntOrdList.add 5
10 |> IntOrdList.add 3
11 |> IntOrdList.add 10
12 |> IntOrdList.to_list
13 - : IntOrdList.elem list = [3; 5; 10]
```

Useful Library Types, Modules and Functors

- Option types: `type 'a option = None | Some of 'a`
- Set modules: obtained through the functor `Set.Make` (requires a module `OrderedType`)

```
1 module ISet = Set.Make(Int)
2
3 ISet.union (ISet.singleton 0) (ISet.singleton 2)
```

- Map modules: obtained through the functor `Map.Make` (requires a module `OrderedType`)

```
1 module SMap = Map.Make(String)
2 let m = SMap.add "ciao" 0 (SMap.singleton "boh" 0)
3
4 # SMap.find_opt "ciao" m;;
5 - : int option = Some 0
6
7 # SMap.find_opt "bao" m;;
8 - : int option = None
```

Exceptions


```
1 5 / 0;;
2 Exception: Division_by_zero.
3
4 # try 5 / 0 with Division_by_zero -> 42;;
5 - : int = 42
6
7 # exception My_exception of string;;
8 exception My_exception of string
9
10 # try if true then raise (My_exception "hello") else 0
11 with My_exception s -> print_endline s; 42;;
12 hello
13 - : int = 42
```


How to manage function call in the compiler?

- Names and associated values are local to the function
- It starts with the global environment and create an activation record for each call
- Activation records are stacked: names are resolved as locally as possible
- Using recursion instead of while loops can cause problem when activation records are
- The famous stack overflow error!

Tail Recursion

```
1 let rec len ls = match ls with
2   | [] -> 0
3   | _::ls -> 1 + (len ls)
4   in lrev [0;1;2]
```



In the end: $1 + (1 + (1 + 0)) = 3$

Tail Recursion : the recursive value is just returned!

```
1 let rec find x ls = match ls with
2   | [] -> false
3   | l::_ when l = x -> true
4   | _::ls' -> find x ls'
5   in find 2 [0;1;2]
```

global

find 2 [0;1;2]
global

find 2 [1;2]
find 2 [0;1;2]
global

find 2 [2]
find 2 [1;2]
find 2 [0;1;2]
global


For each frame: a value for `ls`, for `l`, and return value

In the end: just `true`! **No need to keep the records!**

Tail Recursion : the recursive value is just returned!

```
1 let rec find x ls = match ls with
2   | [] -> false
3   | l::_ when l = x -> true
4   | _::ls' -> find x ls'
5   in find 2 [0;1;2]
```

The compiler is smart enough to compile it into the following:



A lot more efficient (both in space and time)!

Note: library functions are tail recursive (see the documentation)

E.g. `fold_left` is tail recursive, `fold_right` is not!

Tail Recursion though accumulators

A tail recursive len:

```
1 let len ls =  
2   let rec helper ls' n =  
3     match ls' with  
4     | [] -> n  
5     | _::ls'' -> helper ls'' (n + 1)  
6   in helper ls 0
```

Note: we are operating on the parameter, not on the value returned by the function

Tail Recursion in Continuation-Passing-Style

Another tail recursive len:

```
1 let len ls =  
2   let rec helper ls' k =  
3     match ls' with  
4     | [] -> k 0  
5     | _::ls'' -> helper ls'' (fun x -> k x + 1)  
6   in helper ls (fun x -> x)
```

- The function `k` represent how to continue the computation after the recursive call
- A lot less clear then the previous one
- Sometimes it is needed with recursive data structures that are not linear (e.g. trees)

Exercises

Exercise 8. Define a type for Finite State Automata and a function for checking if a given string is inside the generated language. But this time use Maps and Sets!

Exercise 9. Write a tail recursive function for computing the sum of the elements in the leaves of a binary tree of integers.

Exercise 10. Write a module with an abstract type `stack` and with functions for pushing a value on top of the stack (`push n`), one for removing and returning the element on top (`pop`), and one for updating the two top elements by performing an arithmetical operation (`update`). Define a type for programs composed by sequences of pushes and updates, and a function `exec` to get the result of executing it over the empty stack.

Example: `exec [Push 2; Push 3; Push 4; Apply sum]`
returns a stack with 7 on top and then 2.